HRP 259 SAS LAB FIVE  November 19, 2008


Lab Five: Data exploration exercise: linear regression

Lab Objectives

After today’s lab you should be able to:

1. Explore a dataset to identify outliers and missing data.

2. Plot data distributions. 

3. Check for normality.

4. Obtain Pearson’s correlation coefficients between multiple covariates (must be continuous or binary).

5. Build a linear regression model.

6. Dummy code categorical predictors.

7. Use a class statement to make SAS dummy code for you.

8. Look for confounding in the context of linear regression.

9. Walk through a real data analysis exercise.

LAB EXERCISE STEPS:

Follow along with the computer in front…

1. Double-click on the SAS icon to open SAS.

2. Goto the class website: www.stanford.edu/~kcobb/courses/lab5 --> Download Lab 5 DATA, which is already in SAS format. Place the dataset on your desktop.

The dataset contains data on 99 women runners. Variables include

Outcome:
pchange= percent change in spine bone density since baseline 

Predictors:

weightchg=percent change in weight since baseline
coffee = coffee drinking (cups/day)

stddrink = number of standard drinks of alcohol per day

dairy= dairy intake (servings of dairy/day)

kcal= calorie intake (kcal/day)

soda= soda intake (ounces/day)

ed=history of an eating disorder (yes/no)

veggie=vegetarian (yes/no)

weeklymiles = miles run per week

lifting = weight lifting (minutes/week)

The aim of the study is to test whether coffee and/or alcohol affect changes in bone density, controlling for potential confounders and effect modifiers.

3. Use point-and-click (slamming file cabinet icon) to name a library, lab5, that points to your desktop where the SAS dataset is sitting. 

a. Double Click the “New Library” icon (slamming file cabinet) on your menu bar 

b. Type lab5 in the library Name box.

c. Browse to find your desktop. Then click OK.

d. Click OK to exit the new library screen.

4. Examine the distributions of variables in the class dataset using point-and-click as follows:

a. From the menu select: Solutions(Analysis(Interactive Data Analysis
b. Double click to open: library “lab5”, dataset “class”

c. To see distribution of each variable: Analyze(Distribution(Y)
d. To see list and frequency of different values (after distribution screen is already open): Tables(Frequency Counts
e. To test for normality: Tables(Tests for Normality
f. To get a scatter plot of any two variables: Analyze(Fit (Y X)(use point and click to identify predictor variable (X) and outcome variable (Y)
5. Use this point-and-click screen to answer questions such as:

a. How many subjects are in the study?

b. What is the range of outcome values in the study?

c. Is the outcome variable normally distributed?

d. Can you find any outliers or data entry errors in the data?

e. Which predictors appear linearly related to changes in bone density in a basic scatter plot?

6. Close point-and-click screen. Using code, view the correlation matrix between all predictor variables to identify which predictors are correlated. 


proc corr data=lab5.class best=3;


var weightchg kcal dairy soda coffee


stddrink ed veggie  weeklymiles lifting; 

run;

7. Get the correlation matrix between the outcome variable (pchange) and all the predictors:

proc corr data=lab5.class;


var weightchg kcal dairy soda coffee


stddrink ed veggie  weeklymiles lifting;


with pchange;

run;
8. Use an automatic selection procedure to find the strongest predictors, as another way to help explore your data:




proc reg data=lab5.class;

model pchange = weightchg kcal dairy soda coffee


stddrink ed veggie  weeklymiles lifting/selection=stepwise sle=.10;

run;
At the end of your output, find:

Summary of Stepwise Selection

        Variable      Variable      Number    Partial     Model

 Step   Entered       Removed       Vars In   R-Square   R-Square    C(p)     F Value   Pr > F

   1    coffee                          1      0.0707     0.0707     5.4533      6.62   0.0118
   2    weightchg                       2      0.0663     0.1370     0.9998      6.61   0.0119



9. Test the hypothesis that coffee drinking is related to bone density change, starting with a univariate regression:

proc reg data=class;

model pchange = coffee;

run;
Parameter Estimates

                                   Parameter       Standard

              Variable     DF       Estimate          Error    t Value    Pr > |t|

              Intercept     1        0.24478        0.33066       0.74      0.4609

              coffee        1        0.49333        0.17419       2.83      0.0056
Univariate model:

Percent change in spine bone density (%) = 0.25 + 0.50 (cups coffee/day)

Interpretation: on average, women increased just 0.25% in spine bone density during the study; each cup of daily coffee was associated with an additional 0.5% increase. 

In univariate regression, coffee is a significant positive predictor of bone density changes.

Also, visually inspect the scatter plot between coffee and pchange:

proc gplot data=lab5.class;

plot pchange*coffee;

symbol v=dot c=red i=rl;

run;

proc gplot data=lab5.class;

plot pchange*coffee;

symbol v=dot c=red i=sm60s;

run;

This plot reveals some interesting things—The effect of coffee appears to begin after 1 cup and to level off after 3 cups. This suggests another way to model coffee drinking—you could compare 0-1 cup drinkers to 2-3 cup drinkers to 3+ drinkers.

10. Test for confounding by the potential confounders. For example, coffee drinking may be associated with increased dairy intake (e.g., lattes) and increased dairy may explain the link with bone density. 

To test for confounding, include each covariate in the model one at a time with the main predictor and see how inclusion of the covariate affects the relationship between coffee and the outcome (i.e., affects the beta coefficient). To speed up the process of doing this, use a MACRO:


%macro speed(confounder);

proc reg data=lab5.class;

model pchange = coffee &confounder.;

run;

%mend;

%speed(kcal);

%speed(stddrink);

%speed(dairy);

%speed(weightchg);

%speed(soda);

%speed(ed);

%speed(veggie);

%speed(weeklymiles);

%speed(lifting);

Scroll through the output to find that there is little evidence of confounding. The biggest confounder is a history of eating disorders (ed). Including this variable (ed) changes the point estimate of coffee drinking from a beta of 0.49 to a beta of 0.43. As a rule of thumb, changes greater than 10% are considered confounders; we just barely make this cutoff here.

Why is this a confounder? Women with a history of eating disorders tend to drink more coffee and they have a greater increase in bone density in this sample (probably because they start with greatly diminished bone density and are catching up on their bone mineralization). 

11. Now, assemble your final model: include coffee, ed (because of the slight evidence of confounding), and weight change (since weight change is itself a strong predictor).

proc reg data=lab5.class;

model pchange=coffee ed weightchg;

output out=diagnostics rstudent=residual;

run;
FINAL MODEL:

  Parameter Estimates

                                   Parameter       Standard

              Variable     DF       Estimate          Error    t Value    Pr > |t|

              Intercept     1        0.06986        0.35639       0.20      0.8450

              coffee        1        0.48544        0.17459       2.78      0.0066

              ed            1        0.50312        0.59813       0.84      0.4024
              weightchg     1        0.14374        0.04247       3.38      0.0010

Final model: 

Change in spine bone density (%) = 0.06 + 0.485*(cups coffee/day) +0.50*(1 if history of an eating disorder) + 0.14*(% change in weight)

For example, study subject 1 has pchange of 2.5%, weight change of -.29%, coffee of 2.0 cups/day, and ed=1. So, her predicted change in bone density is:

Predicted change in spine bone density (%) = 0.06 + 0.485*(2.0) +0.50*(1) + 0.14*(-.29) = 1.50
Residual = observed-predicted = 2.5-1.5= 1.0

12. Plot the residuals against each predictor to look for patterns:

axis1 label=(angle=90);

proc gplot data=diagnostics;

plot residual*coffee /vaxis=axis1;

symbol v=dot c=red i=sm60s;

run;

13. As I mentioned before, you may also want to try coffee as a categorical predictor, because it looks like the pattern is not strictly linear. To model coffee as categorical, create dummy codes for the different categories:

data class;

set lab5.class;

if 1<coffee<=3 then medcoff=1; else medcoff=0;

if coffee>3 then highcoff=1; else highcoff=0;

run;

proc freq data=class;

tables medcoff highcoff;

run;

The FREQ Procedure

                                                      Cumulative    Cumulative

                  medcoff    Frequency     Percent     Frequency      Percent

                  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                        0          87       87.88            87        87.88

                        1          12       12.12            99       100.00

                                                      Cumulative    Cumulative

                 highcoff    Frequency     Percent     Frequency      Percent

                 ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                        0          87       87.88            87        87.88

                        1          12       12.12            99       100.00


proc reg data=class;

model pchange=medcoff highcoff ed weightchg;

run;
    Parameter Estimates

                                   Parameter       Standard

              Variable     DF       Estimate          Error    t Value    Pr > |t|

              Intercept     1        0.13361        0.35257       0.38      0.7056

              medcoff       1        0.11088        0.85671       0.13      0.8973

              highcoff      1        2.58672        0.86836       2.98      0.0037

              ed            1        0.62527        0.59264       1.06      0.2941

              weightchg     1        0.13668        0.04207       3.25      0.0016
Interpretation: the lowest group increases 0.13% per year (intercept); the medium coffee group increases 0.13+0.11=0.24% per year, but this is not significantly different than the lowest group (p=.89); the highest group increases 0.13+2.59=2.72% per year, which is significantly different than the lowest group (p=.0037). 

14. The other way to do categorical predictors is to let SAS do the dummy coding for you. In PROC GLM, if you put a categorical variable on the class line, SAS will automatically dummy code it:

data class;

set lab5.class;

if coffee<=1 then coffeec="clow";

else if coffee<=3 then coffeec="bmed";

else if coffee>3 then coffeec="ahigh";

run;

proc glm data=class;

class coffeec ;

model pchange=coffeec ed weightchg/solution;

lsmeans coffeec /pdiff adjust=tukey;

run;

             Standard

            Parameter              Estimate             Error    t Value    Pr > |t|


            Intercept           0.133614310 B      0.35256857       0.38      0.7056

            coffeec   ahig      2.586722111 B      0.86835631       2.98      0.0037

            coffeec   bmed      0.110880168 B      0.85671217       0.13      0.8973
            coffeec   clow      0.000000000 B       .                .         .

            ed                  0.625270939        0.59264305       1.06      0.2941

            weightchg           0.136681924        0.04206991       3.25      0.0016

NOTE: The X'X matrix has been found to be singular, and a generalized inverse was used to solve

      the normal equations.  Terms whose estimates are followed by the letter 'B' are not

      uniquely estimable. 

The GLM Procedure

                                      Least Squares Means

                       Adjustment for Multiple Comparisons: Tukey-Kramer

                                              pchange      LSMEAN

                              coffeec          LSMEAN      Number

                              ahig         2.94068176           1

                              bmed         0.46483982           2

                              clow         0.35395965           3

                             Least Squares Means for effect coffeec

                              Pr > |t| for H0: LSMean(i)=LSMean(j)

                                  Dependent Variable: pchange

                         i/j              1             2             3

                            1                      0.0683        0.0102
                            2        0.0683                      0.9908

                            3        0.0102        0.9908

15. If time, repeat procedures for alcohol and bone density: 

proc reg data=class;

model pchange = stddrink;

run;
                                      Parameter Estimates

                                   Parameter       Standard

              Variable     DF       Estimate          Error    t Value    Pr > |t|

              Intercept     1        0.85829        0.36135       2.38      0.0195

              stddrink      1       -0.55462        0.72984      -0.76      0.4492
proc gplot data=lab5.class;

plot pchange*stddrink;

symbol v=dot c=red i=rl;

run;

proc gplot data=lab5.class;

plot pchange*stddrink;

symbol v=dot c=red i=sm60s;

run;

%macro speed(confounder);

proc reg data=lab5.class;

model pchange = stddrink &confounder.;

run;

%mend;

%speed(kcal);

%speed(coffee);

%speed(dairy);

%speed(weightchg);

%speed(soda);

%speed(ed);

%speed(veggie);

%speed(weeklymiles);

%speed(lifting);

Std drink has a beta coefficient of about -.50 and is non-significant in all cases. So, no evidence of confounding.

%macro speed(interacter);

proc glm data=lab5.class;

model pchange = stddrink &interacter. stddrink *&interacter.;

run;

%mend;

%speed(kcal);

%speed(coffee);

%speed(dairy);

%speed(weightchg);

%speed(soda);

%speed(ed);

%speed(veggie);

%speed(weeklymiles);

%speed(lifting);

No evidence of interaction.

16. FINAL model: 

proc reg data=lab5.class;

model pchange=stddrink weightchg;

output out=diagnostics rstudent=residual;

run;
Parameter Estimates

                                   Parameter       Standard

              Variable     DF       Estimate          Error    t Value    Pr > |t|

              Intercept     1        0.84705        0.34651       2.44      0.0163

              stddrink      1       -0.50550        0.70002      -0.72      0.4720

              weightchg     1        0.13605        0.04436       3.07      0.0028

Interpretation: We do not have evidence of an effect of drinking alcohol on changes in bone density in this sample (but recall, alcohol intake was low).







































































































Same model as before.





The high coffee group is significantly different from the low group and borderline (p=.06) from the medium, after adjusting for multiple comparisons 





Least squares means give the mean predicted value for % change in bone density in each group, at the average values of ed and weight change (based on the regression model). 





You may ignore this warning. 





To get p-values for all pairwise comparisons.





In PROC GLM, identify categorical variables on the CLASS line. SAS dummy codes for you.





Create a categorical variable for coffee (three levels). I want the lowest group to be the reference, so it must be alphabetically last.





Leave the lowest group as your reference group.





12 in the high and med groups, a little small…





The lowest group is left as the reference group.





Make sure all your groups are big enough to make reasonable estimates.





That red line can mislead your eye, so use i=sm60s to ask for a curved line, which better captures more subtle patterns.





sm: asks for smoothed curve


60: you can put any number from 01 to 99 to tell SAS how much curvature you want. If you ask for “01” you get a very zig-zagged line; if you ask for “99” you get a straight line. 60 asks for something in between.


s=sort first (this is important! or you will end up with a big mess)








We will do a lot more on graphing later; but, for now, the symbol statement changes the plotting symbol. Here I’m asking SAS to plot with a red dot, and to superimpose a linear regression line (i=rl says “interpolate” or connect the points with a linear regression line).





Rather than writing a PROC REG out for each potential confounder, simplify your code by writing a single macro and then applying it to each potential confounder.





Ask for residuals so you can check the fit of your final model.





rstudent asks for “studendized” residuals (gives a Z value rather than absolute value).








Restricts correlation matrix to the top 3 correlations for each predictor.





Setting the axis allows us to turn the vertical axis label by 90 degrees.





Recall:” &variable. is a macro variable.














PROC REG is the procedure for basic linear regression.  








p-value for the beta-coefficient for each predictor. 


Null hypothesis: beta=0





Partial R-square tells you how much of the variability in the outcome is explained by the predictor variable, after controlling for previously listed predictors. So, alone, coffee drinking explains 7.1% of the variance in bone density changes. After taking into coffee drinking, weight change explains an additional 6.6% of the variance. 





These two variables held up as the strongest predictors of bone density changes in this sample. They are listed in order of their predictive strength.





sle=.10 tells SAS to only enter variables in the model if they have a p-value of <=.10; and remove them from the model if their p-value increases to >.10 when combined with the other variables. 








Asks SAS to use stepwise selection to find the predictors that most strongly predict change in bone density (i.e., smallest p-values). 


**NOTE: Automatic selection are purely exploratory; they should NOT be used for producing the final model, especially when our goal is to test a particular hypothesis.

